
Ejercicios Lección 4. Acceso a Bases de Datos 

 Ejemplo práctico: Control Data Grid 

Queremos hacer un programa de consulta en una base de datos utilizando un control 

DataGrid. 

En diseño tenemos lo siguiente: 

PASOS ACCIÓN 

1 Nuevo proyecto EXE 

2 Añadimos un nuevo componente: el control DataGrid 



3 Añadimos la Referencia ADO 

4 Añadimos un cuadro de texto, label y un botón 

5 Añadimos un control DataGrid 

6 Comenzamos con el código indicando constantes para el valor de la cadena de 
conexión, la consulta sql y el campo por el que se va a filtrar 

Option Explicit 



' Cadena de conexión 
Private Const s_CONNECTION_STRING As String = 
"Provider=Microsoft.Jet.OLEDB.4.0;" & _ 

 "Data Source=C:\Archivos de programa" & _ 
 "\Microsoft Visual Studio\VB98\" & _ 
 "NWIND.MDB;Persist Security Info=False" 

Private Const SQL As String = "Select * From Proveedores" 
Private Const CAMPO_A_FILTRAR As String = "NombreContacto" 
'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''' 

' Variable de tipo Recordset y con evento 
Private WithEvents Recordset As ADODB.Recordset 

7 Añadimos el código cuando se produce un error en el recordset y se dispara este 
evento 

Private Sub Recordset_Error( _ 
 ByVal ErrorNumber As Long, _ 
 Description As String, _ 
 ByVal Scode As Long, _ 
 ByVal Source As String, _ 
 ByVal HelpFile As String, _ 
 ByVal HelpContext As Long, _ 
 fCancelDisplay As Boolean) 

  ' Mostramos el error 
  MsgBox " Descripción del Error :" & Description, vbCritical 
End Sub 

8 Añadimos el código del botón que quita el fitro y vuelve a visualizar 
  todos los datos del recordset en el dataGrid 

Private Sub Command1_Click() 

    ' Comprueba que el recordset está instanciado 
    If Not Recordset Is Nothing Then 

 ' Comprueba que esté abierto 
 If Recordset.State = adStateOpen Then 

 ' Elimina el filtro 
 Recordset.Filter = "" 
 ' Carga el Datagrid 
 Set DataGrid1.DataSource = Recordset 
 ' se posiciona en el primer registro 
 Recordset.MoveFirst 

 End If 
    End If 
End Sub 

9 Añadimos el código para mostrar en el caption del formulario la posición 
 actual del recordset mediante la propiedad AbsolutePosition 

Private Sub Recordset_MoveComplete( _ 
    ByVal adReason As ADODB.EventReasonEnum, _ 
    ByVal pError As ADODB.Error, _ 



 adStatus As ADODB.EventStatusEnum, _ 
    ByVal pRecordset As ADODB.Recordset) 

  Me.Caption = " Registro actual: " & CStr(Recordset.AbsolutePosition) 
End Sub 

10 Añadimos el evento Load del Form 

Private Sub Form_Load() 

    ' Variable para la conexión Ado 
    Dim db As ADODB.Connection 

    ' Nuevo objeto Connection 
    Set db = New ADODB.Connection 

    db.CursorLocation = adUseClient 

    ' Cadena de conexión 
    db.Open s_CONNECTION_STRING 

    ' Nuevo objeto Recordset 
    Set Recordset = New Recordset 

    ' Abre el Recordset con la consulta 
    Recordset.Open SQL, db, adOpenStatic, adLockOptimistic 

    ' Asigna el recordset al datagrid 
    Set DataGrid1.DataSource = Recordset 

    ' Opcional . esto hace que se seleccione la fila completa en el DataGrid 
    DataGrid1.MarqueeStyle = dbgHighlightRowRaiseCell 

    Text1 = "" 
    Command1.Caption = "Quitar Filtro" 
    Me.Caption = " Ejemplo para filtrar datos en un DataGrid con Ado " 

End Sub 

11 Escribimos cuando se filtra los datos del control dataGrid al escribir en Text1 

Private Sub Text1_Change() 

    ' Comprueba que el recordset se halla creado 
    If Recordset Is Nothing Then 

 MsgBox " No se ha creado el recordset", vbCritical 
    Exit Sub 

    End If 

    ' verifica que el recordset se encuentre abierto 
    If Not Recordset.State = adStateOpen Then 

 MsgBox " El recordset no se encuentra abierto", vbCritical 
 Exit Sub 



 End If 

    If Text1 <> "" Then 
 'Fltra por el campo especificado en la constante CAMPO_A_FILTRAR _ 
 y tomando como valor el dato del textbox 

     Recordset.Filter = CAMPO_A_FILTRAR & " LIKE '*" + Text1.Text + "*'" 
    Else 

 ' Si el textbox no tiene nada, ... se limpia el Filtro 
 Recordset.Filter = "" 

 ' Vuelve a mostrar todos los registros en el dataGRid 
 Set DataGrid1.DataSource = Recordset 

 ' Opcional . Mueve el recordset al primer registro 
     Recordset.MoveFirst 

    End If 

End Sub 

12 Y terminamos cuando cierra el recordset y Descarga la referencia 

Private Sub Form_Unload(Cancel As Integer) 

    ' Cierra 
    If Recordset.State = adStateOpen Then 

     Recordset.Close 
    End If 

    ' descarga 
    If Not Recordset Is Nothing Then 

 Set Recordset = Nothing 
    End If 
End Sub 

12 Probamos 




